<commit_before>{ stdenv
, dpkg
, lib
, autoPatchelfHook
, fetchurl
, gtk3
, glib
, desktop-file-utils
, alsa-lib
, libjack2
, harfbuzz
, fribidi
, pango
, freetype
, curl
}:

stdenv.mkDerivation rec {
  pname = "tonelib-gfx";
  version = "4.6.6";

  src = fetchurl {
    url = "https://www.tonelib.net/download/0509/ToneLib-GFX-amd64.deb";
    sha256 = "sha256-wdX3SQSr0IZHsTUl+1Y0iETme3gTyryexhZ/9XHkGeo=";
  };

  buildInputs = [
    dpkg
    gtk3
    glib
    desktop-file-utils
    alsa-lib
    libjack2
    harfbuzz
    fribidi
    pango
    freetype
  ];

  nativeBuildInputs = [
    autoPatchelfHook
  ];

  unpackPhase = ''
    mkdir -p $TMP/ $out/
    dpkg -x $src $TMP
  '';

  installPhase = ''
    cp -R $TMP/usr/* $out/
    mv $out/bin/ToneLib-GFX $out/bin/tonelib-gfx
  '';

  runtimeDependencies = [
    (lib.getLib curl)
  ];

  meta = with lib; {
    description = "Tonelib GFX is an amp and effects modeling software for electric guitar and bass.";
    homepage = "https://tonelib.net/";
    license = licenses.unfree;
    maintainers = with maintainers; [ dan4ik605743 ];
    platforms = platforms.linux;
  };
}<commit_msg>tonelib-gfx: 4.6.6 -> 4.7.0 (#142259)<commit_after>{ lib
, stdenv
, fetchurl
, autoPatchelfHook
, dpkg
, alsa-lib
, freetype
, libglvnd
, curl
, libXcursor
, libXinerama
, libXrandr
, libXrender
, libjack2
}:

stdenv.mkDerivation rec {
  pname = "tonelib-gfx";
  version = "4.7.0";

  src = fetchurl {
    url = "https://www.tonelib.net/download/0930/ToneLib-GFX-amd64.deb";
    hash = "sha256-BcbX0dz94B4mj6QeQsnuZmwXAaXH+yJjnrUPgEYVqkU=";
  };

  nativeBuildInputs = [ autoPatchelfHook dpkg ];

  buildInputs = [
    stdenv.cc.cc.lib
    alsa-lib
    freetype
    libglvnd
  ] ++ runtimeDependencies;

  runtimeDependencies = map lib.getLib [
    curl
    libXcursor
    libXinerama
    libXrandr
    libXrender
    libjack2
  ];

  unpackCmd = "dpkg -x $curSrc source";

  installPhase = ''
    mv usr $out
    substituteInPlace $out/share/applications/ToneLib-GFX.desktop --replace /usr/ $out/
 '';

  meta = with lib; {
    description = "Tonelib GFX is an amp and effects modeling software for electric guitar and bass.";
    homepage = "https://tonelib.net/";
    license = licenses.unfree;
    maintainers = with maintainers; [ dan4ik605743 orivej ];
    platforms = [ "x86_64-linux" ];
  };
}
{'fraction_non_alphanumeric': 0.12633376013657704, 'fraction_numerical': 0.03158344003414426, 'mean_word_length': 2.8426229508196723, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 4, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 2, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2882714', 'n_tokens_mistral': 988, 'n_tokens_neox': 939, 'n_words': 180}